
Abstract

In this article we formalize the Ascoli-Arzelà’s theorem in Mizar. First, we gave

definitions of equicontinuousness and equiboundedness of a set of continuous functions.

Next, we formalized the Ascoli-Arzelà’s theorem using those definitions, and proved this

theorem.

1 はじめに
Ascoli–Arzelà の定理は,連続関数の族の一様位相に関する相対コンパクト性の必要十

分条件を与える定理である [1] [2] [3] [4] [5]．有界閉区間上の実数値連続関数の集合の元
の任意の列が一様収束する部分列を持つ (点列コンパクト sequentially compactであ
る) ための必要十分条件が同程度連続性の概念を用いて Ascoli と Arzelàによって証明さ
れた. この定理は常微分方程式論におけるペアノの存在定理など多くの定理の証明に用い
られてきた. Fréchetにより定義域のコンパクト距離空間である実数値連続関数の族への
一般化がなされた後,現在は定義域はコンパクトなハウスドルフ空間,値域は任意の距離
空間にまで拡張されている.

筆者らは距離空間から生成されるコンパクトな位相空間からノルム空間から生成され
る位相線形空間への連続関数の集合についての一様位相に関する相対コンパクト性の必
要十分条件に関しての Ascoli–Arzelà の定理とその証明を形式化した. 値域を一般的な距

離空間ではなく,ノルム空間から生成される位相線形空間としたのは筆者らのニューラル
ネットの形式化についての報告 [6]で述べたように,相対コンパクト空間に含まれる関数
の値の代数的演算やノルムを扱う場合の利便性を考慮したためである．本稿ではこれにつ
いて報告する. その形式化はMizar Mathematical Library (以下MML) に ASCOLI [7]

として収録されている. 前回の報告 [8]で,一部その概要に言及したが,その後,内容の修
正その他を行ったので今回,改めてその詳細を報告する.

2 形式化の概要
筆者らの形式化の最終的な結果は以下の通りである.

Listing 1. ASCOLI:23

theorem :: ASCOLI:23
for M be non empty MetrSpace,S be non empty compact TopSpace,

T be NormedLinearTopSpace, U be compact Subset of T,
F be non empty Subset of R NormSpace of ContinuousFunctions(S,T),
G be Subset of Funcs(the carrier of M, the carrier of T)

st S = TopSpaceMetr(M) & T is complete & G = F & for f be Function st f in F holds rng f c= U
holds

( Cl(F) is compact implies G is equibounded & G is equicontinuous )
&

( G is equicontinuous implies Cl(F) is compact );

Sは距離空間Mから生成されるコンパクトな位相空間であり, Tはノルム空間から生
成される位相線形空間である.この位相線形空間については [8]で解説したので,詳しくは
これを参照されたい. Tはノルム空間であると同時に位相空間である.

Tには完備性を仮定する.

R NormSpace of ContinuousFunctions(S,T)

は SからTへの連続写像全体の集合が造るノルム空間であり,Tの完備性からこの空間も
完備である.以後,このノルム空間をMMLの記述を引用する場合を除き L(S,T) で表す.

Funcs(the carrier of M, the carrier of T)

はMの元の集合から Tの元の集合への写像全部の集合を表す. MMLでは [9]と同様に
集合論を基礎とする形式化を行っている．群や環，位相空間など数学的な概念は集合と写
像などから構成される構造体 (struct)と，それの特定の性質を表す属性 (attribute)を
定義することによって形式化される．
距離空間の構造体はMETRIC 1 [10]で以下のように形式化されている．

Listing 2. METRIC 1

definition
struct(1−sorted) MetrStruct
(# carrier −> set, distance −>

Function of [:the carrier,the carrier:],REAL #);
end;

carrier はMの点全ての集合を表し, distanceはMの任意の 2点間の距離を与える
[: the carrier, the carrier :]から Rへの写像である. [: the carrier, the carrier :]は
carrierの 2重直積である.

この構造体に距離 distanceについての以下の対称性や三角不等式などの属性が定義
される.

Listing 3. METRIC 1:def 2 ∼ def 5

definition
let A be set;
let f be PartFunc of [:A,A:], REAL;
attr f is Reflexive means :: METRIC 1:def 2

for a being Element of A holds f.(a,a) = 0;
attr f is discerning means :: METRIC 1:def 3

for a, b being Element of A st f.(a,b) = 0 holds a = b;
attr f is symmetric means :: METRIC 1:def 4
for a, b being Element of A holds f.(a,b) = f.(b,a);

attr f is triangle means :: METRIC 1:def 5
for a, b, c being Element of A holds f.(a,c) <= f.(a,b) + f.(b,c);

end;

距離空間は構造体METRIC 1で上記の属性を持った変数型MetrSpaceとして以下
のように導入されている.

Listing 4. METRIC 1

definition
mode MetrSpace is Reflexive discerning symmetric triangle MetrStruct;

end;

位相空間TopSpaceや位相線形空間NormedLinearTopSpaceも同様に構造体 (struct)

と，それの特定の性質を表す属性 (attribute)を定義することによって形式化されている.

Cl(F)は位相空間 Sの部分集合 Fの閉包である．PRE TOPC [11]で以下のように定
義されている.

Listing 5. PRE TOPC:def 7

definition
let S be TopStruct, F be Subset of S;
func Cl F −> Subset of S means :: PRE TOPC:def 7
for p being set st p in the carrier of S holds p in it

iff for G being Subset of S st G is open holds p in G implies F meets G;
projectivity;

end;

構造体とその構成要素である carrierは厳密に区別される.

以後,表現の煩さを避けるため距離空間Mの点の集合 the carrier of M から位相空
間Tの点の集合 the carrier of Mへの写像 を距離空間Mから位相空間Tへの写像と
表現する.

Gが距離空間Mからノルム空間 Tへの写像全部の集合

Funcs(the carrier of M, the carrier of T)

の部分集合とするとき,同程度有界性 equiboundedと同程度連続性 equicontinuous [1]

[2] [4] [5] を以下のように定義する.

Listing 6. ASCORI:def 2 def 4

definition
let M be non empty MetrSpace ,T be RealNormSpace;
let G be Subset of Funcs(the carrier of M,the carrier of T);
attr G is equibounded means :: ASCORI:def 2
ex K be Real st for f be Function of the carrier of M,the carrier of T

st f in G holds for x be Element of M holds ||.f.x.|| <= K ;
end;

definition
let M be non empty MetrSpace,T be RealNormSpace;
let G be Subset of Funcs(the carrier of S,the carrier of T);
pred G is equicontinuous means :: ASCORI:def 4
for e be Real st 0 < e ex d be Real st 0 < d

&
for f be Function of the carrier of M,the carrier of T

st f in G holds for x1,x2 be Point of M st dist(x1,x2) < d holds ||.f.x1−f.x2.|| < e;
end;

冒頭に示した 定理 ASCOLI:23は同程度有界性 equiboundedと同程度連続性
equicontinuousがノルム空間から生成される位相線形空間への連続関数の集合G が一
様位相に関する相対コンパクトであることの必要十分条件であることを表している. 次章
以降,これについて解説する.

3 補題と定義
前章で述べたようにMMLでは位相空間など数学的な概念は集合と写像などから構成

される構造体 (struct)と，それの特定の性質を表す属性 (attribute)を定義することに
よって形式化される．構造体には基となる構造を拡張した構造など上位，下位の階層関係
をもつものもあるが，数学的には上位,下位の階層関係にあっても開発段階にある現状の
MMLではそのように関係付けられていないものがある. その場合,一方の構造体に関し
て形式化されMMLに収録されているものであっても, 数学的には階層関係があるのにも
関わらず,そのまま引用できないものがある.

Mizarは変数の型や構造の違いを厳密に区別する. これは,人手に依らず,計算機の記
号列の処理によって機械的,かつ厳密に定理証明の検証を行うシステムの目的から止むを
得ないことでもある.

例えば,計算機による数値計算のプログラムで,同じ演算子であっても変数型が実数と
整数で,小数点以下を扱うか扱わないかで,結果が異なる場合があり,関数など使用する引
数の型が異なれば,エラーメッセージを出力するなどと同じである.

閉包は位相空間で定義され,距離空間で未定義であるため, 利便性のため以下のように
距離空間での閉包を定義した.

Listing 7. ASCOLI:def 1

definition
let X be non empty MetrSpace, Y be Subset of X;
func Cl Y −> Subset of X means :: ASCOLI:def 1
ex Z be Subset of TopSpaceMetr X st Z = Y & it = Cl Z;

end;

theorem :: ASCOLI:1
for X be RealNormSpace, Y be Subset of X, Z be Subset of MetricSpaceNorm X

MMA-WiP (2022), vol.4, no.4 - https://mizar-jp.org/journal/MMA-WiP/j2022.4.4 - ISSN 2434-5458 (Online) 4/15

https://mizar-jp.org/journal/MMA-WiP/j2022.4.4


Mechanized Mathematics and Its Applications, Works in Progress Ascoli-Arzelà’s Theorem (Miyajima and Yamazaki)

st Y = Z holds Cl(Y) = Cl(Z);

registration
let X be non empty MetrSpace;
let H be non empty Subset of X;
cluster Cl(H) −> non empty;

end;

また, 数学的にはノルム空間は,距離空間であり,距離空間は位相空間である．一方,ノ
ルム空間は線形空間であり, 位相線形空間でもある．報告で解説したように C0SP3 [12]

にはこのノルム空間を基にし位相線形空間の構造を併せ持つNormedLinearTopSpace

を導入したが, さらに利便性を高めるため関数の連続性に関しての以下の補題を用意した.

Listing 8. ASCOLI:3

theorem :: ASCOLI:3
for S be non empty TopSpace,T be NormedLinearTopSpace,

f be Function of S,T, x be Point of S holds
( f is continuous at x iff for e be Real st 0 < e holds ex H being Subset of S st
( H is open & x in H & for y be Point of S st y in H holds ||.f.x−f.y.|| < e ) );

theorem :: ASCOLI:4
for S be non empty MetrSpace, V be non empty compact TopSpace,

T be NormedLinearTopSpace, f be Function of V,T
st V = TopSpaceMetr(S)

holds
f is continuous

iff for e be Real st 0 < e holds ex d be Real st 0 < d
& for x1,x2 be Point of S st dist(x1,x2) < d holds ||.f/.x1−f/.x2.|| < e;

距離空間には TBSP 1 [13]で点列の収束 convergence, 全有界性 totally bounded

や完備性 completeなどの概念が定義されている．

Listing 9. TBSP 1

definition
let N;
attr N is totally bounded means :: TBSP 1:def 1
for r st r>0 ex G st G is finite & the carrier of N = union G & for C st C in G ex w st C = Ball(w,r);

end;

reserve S1 for sequence of M,
S2 for sequence of N;

theorem :: TBSP 1:4
f is sequence of N iff dom f = NAT & for n holds f.n is Element of N;

definition
let N,S2;
attr S2 is convergent means :: TBSP 1:def 2
ex x being Element of N st for r st r>0
ex n st for m st n<=m holds dist(S2.m,x)<r;

end;

definition
let M,S1;
assume
S1 is convergent;
func lim S1 −> Element of M means :: TBSP 1:def 3
for r st r>0 ex n st for m st m>=n holds dist(S1.m,it)<r;

end;

definition
let N,S2;
attr S2 is Cauchy means :: TBSP 1:def 4

MMA-WiP (2022), vol.4, no.4 - https://mizar-jp.org/journal/MMA-WiP/j2022.4.4 - ISSN 2434-5458 (Online) 5/15

https://mizar-jp.org/journal/MMA-WiP/j2022.4.4


Mechanized Mathematics and Its Applications, Works in Progress Ascoli-Arzelà’s Theorem (Miyajima and Yamazaki)

for r st r>0 ex p st for n,m st p<=n & p<=m holds dist(S2.n,S2.m)<r;
end;

definition
let N;
attr N is complete means :: TBSP 1:def 5
for S2 st S2 is Cauchy holds S2 is convergent;

end;

ノルム空間 Tは 周知のように距離 distanceを

distance.(x,y) = ||.x− y.||

で定義すれば距離空間にもなる.NORMSP 2 [14]でこれをMetricSpaceNorm Tで表
し定義している.

MetricSpaceNorm Tの距離の作り方から T の完備性は Tから生成される距離空
間MetricSpaceNorm Tが完備であることと同値であり,以下の命題が成り立つ.

Listing 10. ASCOLI:7

theorem :: ASCOLI:7
for T be RealNormSpace, H be non empty Subset of MetricSpaceNorm T
st T is complete holds (MetricSpaceNorm T) | Cl(H) is complete;

また全有界性 totally boundedと閉包の定義から以下も成り立っている.

Listing 11. ASCOLI:8

theorem :: ASCOLI:8
for T be RealNormSpace,

H be non empty Subset of MetricSpaceNorm T
holds

(MetricSpaceNorm T) | H is totally bounded
iff (MetricSpaceNorm T) | Cl(H) is totally bounded;

以上の準備のもとに,完備なノルム空間 T から生成される完備距離空間
MetricSpaceNorm Tについて, その部分集合Hの全有界性 totally bounded, 閉包
Cl(H)の点列コンパクト性 sequentially compact, 相対コンパクト性が何れも同値で
あること示す以下の命題を形式化した.

Listing 12. ASCOLI:10

theorem :: ASCOLI:10
for Z be RealNormSpace, F be non empty Subset of Z,

H be non empty Subset of MetricSpaceNorm Z,
T being Subset of TopSpaceNorm Z st Z is complete & H = F & H = T

holds
( (MetricSpaceNorm Z) | H is 　 totally boundediff Cl(H) is sequentially compact) &
( (MetricSpaceNorm Z) | H is 　 totally boundediff (MetricSpaceNorm Z) | Cl(H) is compact) &
( (MetricSpaceNorm Z) | H is 　 totally boundediff Cl(F) is compact) &
( (MetricSpaceNorm Z) | H is 　 totally boundediff Cl(T) is compact);

4 Ascoli–Arzelà の定理
周知のように距離空間Mは開球 Ballを用いて開集合の族を定義でき,位相空間を構

成できる． PCOMPS 1 [15] にはこれを以下のように形式化している.

Listing 13. PCOMPS 1

definition
let M be MetrStruct;
func Family open set(M) −> Subset−Family of M means :: PCOMPS 1:def 4
for V holds V in it iff for x st x in V holds ex r st r>0 & Ball(x,r) c= V;

end;
theorem :: PCOMPS 1:33
TopStruct (#the carrier of M,Family open set(M)#) is TopSpace;

definition
let M be MetrStruct;
func TopSpaceMetr M −> TopStruct equals :: PCOMPS 1:def 5
TopStruct (#the carrier of M, Family open set(M)#);

end;

Family open set(M) が開球による開集合の属であり, TopSpaceMetr M がそれ
によって構成される位相空間である.

さらに本稿の形式化では以下の命題を用いている.

Listing 14. TBSP 1:8，TBSP 1:9

theorem :: TBSP 1:8
TopSpaceMetr(T) is compact implies T is complete;

theorem :: TBSP 1:9
N is Reflexive triangle & TopSpaceMetr(N)

is compact implies N is totally bounded;

これにより,距離空間M が完備 completeならばMから構成される．
位相空間 TopSpaceMetr Mも完備であり, TopSpaceMetr Mが compactであ

れば,Mは全有界 totally boundedである.

これと,前章で述べた補題を用いてAscoli–Arzelà の定理の形式化を以下の順に行った.

以下,T の完備性を仮定する. T の完備性により,S から T への連続写像全体の集
合が造るノルム空間 L(S,T) は完備である. このノルム空間から構成される距離空間
MetricSpaceNorm L(S,T) も完備である. Hがこの空間の部分集合のとき,H の閉包
Cl(H) の列コンパクト性 sequentially compact と全有界性 totally boundedの同値
性についての以下の命題とその証明を形式化した.

Listing 15. ASCOLI:11，ASCOLI:12

theorem :: ASCOLI:11
for S be non empty compact TopSpace,T be NormedLinearTopSpace
st T is complete holds

for H be non empty Subset of
(MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T))

holds
Cl(H) is sequentially compact iff
(MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T) )

| H is totally bounded;

theorem :: ASCOLI:12
for S be non empty compact TopSpace,T be NormedLinearTopSpace
st T is complete holds
for F be non empty Subset of

R NormSpace of ContinuousFunctions(S,T),
H be non empty Subset of

MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T) st
H = F holds Cl(F) is compact iff
(MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T) )
| H is totally bounded;

MetricSpaceNorm L(S,T) の部分集合Gが全有界 totally boundedならGの元
の有限集合 L と半径 1の開球の有限集合 {B(h, 1) : h ∈ L} が存在して

G ⊆
∪
h∈L

B(h, 1)

となる. ここで
d = max{||h|| : h ∈ L}

とおくと任意の f ∈ G, x ∈ Sについて,

f ∈ G ⊆
∪
h∈L

B(h, 1)

から
||h− f || ≤ 1

となる h ∈ L が存在する．
||h− f || ≤ 1 から任意の x ∈ Sについて,

||f(x) − h(x)|| ≤ 1

であり,

||f(x)|| ≤ ||h(x)|| + ||f(x) − h(x)|| ≤ d + 1

が成り立つことから，前章の定義によりG は同程度有界 equiboundedである.

次に,任意の正の実数 0 < ϵ について上と同様にGの元の有限集合 L と半径 ϵ/3の開
球の有限集合 {B(h, ϵ/3) : h ∈ L} を

G ⊆
∪
h∈L

B(h, ϵ/3)

となるように選ぶ.

各 h ∈ L ⊆ G はMetricSpaceNorm L(S,T) に属する連続写像であり,特に Sの
compact性から一様連続であるから 0 < dh が存在して, Sの任意の 2つの元 x, y ∈ S に
ついて dist(x, y) < dh なら

||h(x) − h(y)|| < ϵ/3

が成り立っている.(前章の命題 ASCOLI:4そのものである)

そこで
d = min{dh : h ∈ L}

とおけば, Gの任意の元 f ∈ Gについて

f ∈ G ⊆
∪
h∈L

B(h, ϵ/3)

から
||h− f || < ϵ/3

となる h ∈ L が存在し

Sの任意の 2つの元 x, y ∈ S について dist(x, y) < d なら dist(x, y) < dh でもある
から

||f(x) − f(y)|| ≤ ||f(x) − h(x)|| + ||h(x) − h(y)|| + ||h(y) − f(y)|| < ϵ/3 + ϵ/3 + ϵ/3 = ϵ

が成り立つことから前章の定義によりG は同程度連続 equicontinuousである. 以上を
形式化したのが次の命題である.

Listing 16. ASCOLI:13

theorem :: ASCOLI:13
for M be non empty MetrSpace,S be non empty compact TopSpace,

T be NormedLinearTopSpace
st S = TopSpaceMetr(M) & T is complete holds
for G be Subset of Funcs(the carrier of M, the carrier of T),

H be non empty Subset of
MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T) st
G = H & (MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T))
| H is totally bounded holds

G is equibounded & G is equicontinuous;

特に,上記の命題で,Tが有限次元空間ならば下に述べる命題 ASCOLI:16の系として,

同程度有界性 equibounded ,連続性 equicontinuousは全有界性 totally boundedの
必要十分条件になり以下の命題が成り立つ. [6]

Listing 17. NEURONS1:26

theorem :: NEURONS1:26
for M be non empty MetrSpace,S be non empty compact TopSpace,
T be NormedLinearTopSpace
st S = TopSpaceMetr(M) & T is complete & T is finite−dimensional
& dim (T) <> 0 holds
for G be Subset of Funcs(the carrier of M, the carrier of T),

H be non empty Subset of
(MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T))

st G = H
holds
(MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T))
| H is totally bounded iff G is equibounded & G is equicontinuous;

また,完備な距離空間MetricSpaceNorm L(S,T)の部分集合の全有界性 totally bounded

と相対 compact性は同値なので以下の命題が成り立つ

Listing 18. NEURONS1:28

theorem :: NEURONS1:28
for M be non empty MetrSpace,S be non empty compact TopSpace,
T be NormedLinearTopSpace
st S = TopSpaceMetr(M) & T is complete & T is finite−dimensional
& dim (T) <> 0 holds

for G be Subset of Funcs(the carrier of M, the carrier of T),
F be non empty Subset of R NormSpace of ContinuousFunctions(S,T)

st G = F
holds Cl(F) is compact iff G is equibounded & G is equicontinuous;

しかしながら,本稿では Tが有限次元空間という制約を課していない. 代わりに以下
の必要十分条件 ASCOLI:16の命題が成り立つ. この命題が本稿報告の形式化の中心的命
題である. 以下のその証明の概略を述べる.

MMA-WiP (2022), vol.4, no.4 - https://mizar-jp.org/journal/MMA-WiP/j2022.4.4 - ISSN 2434-5458 (Online) 9/15

https://mizar-jp.org/journal/MMA-WiP/j2022.4.4


Mechanized Mathematics and Its Applications, Works in Progress Ascoli-Arzelà’s Theorem (Miyajima and Yamazaki)

MetricSpaceNorm L(S,T) の部分集合 G が全有界 totally bounded なら AS-

COLI:13からGは同程度連続 equicontinuous である.

次に x ∈ Sを S の任意元 とする. 任意の正の実数 0 < ϵ について上と同様にGの元
の有限集合 L と半径 ϵの開球の有限集合 {B(h, ϵ) : h ∈ L} を

G ⊆
∪
h∈L

B(h, ϵ)

となるように選ぶ.

これを用いて T の開球の有限集合

{B(h(x), ϵ) : h ∈ L}

をつくると,

{f(x) : f ∈ G} ⊆
∪ ∪

h∈L

B(h.x, ϵ)

が成り立っている. T の部分集合 {f(x) : f ∈ G} は全有界 totally boundedであり,そ
の閉包は compactである.

逆にGが同程度連続 equicontinuousであり,Sの任意の元xについて {f(x) : f ∈ G}
の閉包が compactであるとする.

任意の正の実数 0 < ϵ について, Gの同程度連続性から正の実数 0 < d が存在し, Sの
任意の 2元 x, y ∈ S について

dist(x, y) < dならば ||f(x) − f(y)|| < ϵ/8 (∀f ∈ G) (1)

が成り立っている
そこで Sの開球の集合 {Ball(x, d) : x ∈ S} をつくると

S ⊆
∪
x∈S

Ball(x, d)

であるから Sの開被覆である.

Sは compactであるから Sの元の有限列 {xi}1≤i≤n が存在して

S ⊆
∪

1≤i≤n

Ball(xi, d)

が成り立つ.

この {xi}1≤i≤n を用いて各 iについて {f(xi) : f ∈ G}の閉包 Cl{f(xi) : f ∈ G}を
その要素とする T の部分集合の有限列

{Cl{f(xi) : f ∈ G}}1≤i≤n

をつくる. 各Cl{f(xi) : f ∈ G}は仮定により compactであるからこの部分集合列の和

B =
∪

1≤i≤n

Cl{f(xi) : f ∈ G}

も compact集合である.

この Tの compact部分集合 B を覆う B の元の有限列とその各点を中心とする半径
ϵ/8 開球の有限列

{wj} 1≤j≤m, {Ball(wj , ϵ/8)} 1≤j≤m,

B ⊆
∪

1≤j≤m

Ball(wj , ϵ/8) (2)

を構成する. この開球の有限列を用いて以下のようにしてmn個のGを覆う開球を構成す
る. 自然数の集合 {i ∈ N : 1 ≤ i ≤ n}から集合 {j ∈ N : 1 ≤ j ≤ m} への写像全体の集合

{σ : {i ∈ N : 1 ≤ i ≤ n} → {j ∈ N : 1 ≤ j ≤ m}}

を P で表すと,この有限集合の要素の数はmn 個である. σ ∈ P に対してGの部分集合
を対応させる写像 K

K(σ) = {f ∈ G : (∀i : 1 ≤ i ≤ n)(f(xi) ∈ Ball(wσi
, ϵ/8)}

を導入する.

任意の σ ∈ P に対してGの任意の 2 元 f, g について f, g ∈ K(σ) ならば

f(xi), g(xi) ∈ Ball(wσi
, ϵ/8) (∀i : 1 ≤ i ≤ n)

から

||f(xi) − g(xi)|| ≤ ||f(xi) − wσi
|| + ||wσi

− g(xi)|| < ϵ/8 + ϵ/8 = ϵ/2 (∀i : 1 ≤ i ≤ n)

を得る.

そこで x ∈ S を任意に選ぶと,

S ⊆
∪

1≤i≤n

Ball(xi, d)

から 1 ≤ i ≤ nが存在して x ∈ Ball(xi, d) となり,これから dist(xi, x) < d である. (1)

式から
||f(xi) − f(x)|| < ϵ/8, ||g(xi) − g(x)|| < ϵ/8

であり,

||f(x) − g(x)|| ≤ ||f(x) − f(xi)|| + ||f(xi) − g(xi)|| + ||g(xi) − g(x)||

< ϵ/8 + ϵ/8 + ϵ/8

< ϵ/2

結局

||f − g|| < ϵ/2　 (∀f, g ∈ K(σ)) (3)

Gの任意の元 f について (2)式 から任意の 1 ≤ i ≤ nに対して 1 ≤ j ≤ mが存在して

f(xi) ∈ Ball(wj , ϵ/8)

従って,Gの任意の元 f について σ ∈ P が存在して

f ∈ K(σ) (4)

ここで, 任意の σ ∈ P について, K(σ) ̸= ϕ ならば Fσ ∈ K(σ) である Fσ ∈ G が存在する.

g ∈ K(σ) である任意の g ∈ G については (3)式 から

||Fσ − g|| < ϵ/2

が得られ
g ∈ Ball(Fσ, ϵ)

従って
K(σ) ⊆ Ball(Fσ, ϵ) (5)

Gの開球の有限列 {Ball(Fσ, ϵ)}σ∈P は (5)式 から

G ⊆
∪
σ∈P

Ball(Fσ, ϵ)

よってGは全有界 totally bounded である.

以上から次の命題の形式証明を得た.

Listing 19. ASCOLI:16

theorem :: ASCOLI:16
for M be non empty MetrSpace,S be non empty compact TopSpace,

T be NormedLinearTopSpace,
G be Subset of Funcs(the carrier of M, the carrier of T),
H be non empty Subset of
MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T)

st S = TopSpaceMetr(M) & T is complete & G = H holds
(MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T))

| H is totally bounded
iff
G is equicontinuous &
for x be Point of S,

Hx be non empty Subset of MetricSpaceNorm T
st Hx = {f.x where f is Function of S,T :f in H }

holds (MetricSpaceNorm T) | Cl(Hx) is compact;

これは完備ノルム空間L(S,T)から生成される完備距離空間MetricSpaceNorm L(S,T)

の部分集合の全有界性 totally boundedと同程度連続性 equicontinuous,

MetricSpaceNorm L(S,T) に属する関数の各 Sの点 xに対する値の集合の閉包の
compact性が同値であること示している.

さらに完備距離空間の部分集合の全有界性 totally boundedと閉包の点列コンパク
ト性，相対コンパクト性は何れも同値であるから以下の命題が成り立つ.

Listing 20. ASCOLI:17，ASCOLI:18

theorem :: ASCOLI:17
for M be non empty MetrSpace,S be non empty compact TopSpace,

T be NormedLinearTopSpace,
G be Subset of Funcs(the carrier of M, the carrier of T),
H be non empty Subset of
(MetricSpaceNorm R NormSpace of ContinuousFunctions(S,T))

st S = TopSpaceMetr(M) & T is complete & G = H
holds

Cl(H) is sequentially compact
iff G is equicontinuous

&
for x be Point of S, Hx be non empty Subset of MetricSpaceNorm T
st Hx = {f.x where f is Function of S,T :f in H }

holds (MetricSpaceNorm T) | Cl(Hx) is compact;

theorem :: ASCOLI:18
for M be non empty MetrSpace,S be non empty compact TopSpace,

T be NormedLinearTopSpace,
F be non empty Subset of
R NormSpace of ContinuousFunctions(S,T),

G be Subset of Funcs(the carrier of M, the carrier of T)
st S = TopSpaceMetr(M) & T is complete & G = F holds

Cl(F) is compact
iff
G is equicontinuous

& for x be Point of S,
Fx be non empty Subset of MetricSpaceNorm T
st Fx = {f.x where f is Function of S,T :f in F }
holds

(MetricSpaceNorm T) | Cl(Fx) is compact;

また,証明の簡潔化に必要な補助的な命題としてTのcompact性がTopSpaceNormT

の compact性 と同値であること, T の compact性 からTopSpaceNorm Tの完備性
completeが従うことを表す以下の命題とその証明を形式化した.

Listing 21. ASCOLI:20 ∼ ASCOLI:22

theorem :: ASCOLI:20
for T be NormedLinearTopSpace holds
T is compact iff TopSpaceNorm T is compact;

theorem :: ASCOLI:21
for T be NormedLinearTopSpace, X be set holds
X is compact Subset of T iff X is compact Subset of TopSpaceNorm T;

theorem :: ASCOLI:22
for T be NormedLinearTopSpace st T is compact holds T is complete;

以上の準備のもとに最後に本稿の目的である定理 ASCOLI:23の証明を形式化した.

Listing 22. ASCOLI:23

theorem :: ASCOLI:23
for M be non empty MetrSpace,S be non empty compact TopSpace,

T be NormedLinearTopSpace,
U be compact Subset of T,
F be non empty Subset of
R NormSpace of ContinuousFunctions(S,T),
G be Subset of Funcs(the carrier of M, the carrier of T)

st S = TopSpaceMetr(M) & T is complete
& G = F & for f be Function st f in F holds rng f c= U

holds
( Cl(F) is compact implies G is equibounded & G is equicontinuous )

&
( G is equicontinuous implies Cl(F) is compact );

5 まとめ
本稿では距離空間から生成されるコンパクトな位相空間からノルム空間から生成され

る位相線形空間への連続関数の集合についての一様位相に関する相対コンパクト性の必

要十分条件に関しての Ascoli–Arzelà の定理とその証明の形式化を報告した. 関数の値域
を一般的な距離空間ではなく,ノルム空間から生成される位相線形空間としたのは筆者ら
のニューラルネットの形式化についての報告 [6]で述べたように, 相対コンパクト空間に
含まれる関数の値の代数的演算やノルムを扱うためである．今後,この形式化を用いて,

ニューラルネットによる非線形関数の近似定理の形式化を行う.

